```
> restart:with(LinearAlgebra):
```




```
> o0:=<0,0,0>;
```

$$o0 := \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix} \tag{1}$$

```
> Mrotztransl:=(alpha,point)->&lt;&lt;cos(alpha),sin(alpha),0,0&gt;|&lt;-sin(alpha),
  cos(alpha),0,0&gt;|&lt;0,0,1,0&gt;|&lt;point[1],point[2],point[3],1&gt;&gt;;
```

$$Mrotztransl := (\alpha, point) \mapsto \left\langle \langle \cos(\alpha), \sin(\alpha), 0, 0 \rangle \middle| \langle -\sin(\alpha), \cos(\alpha), 0, 0 \rangle \middle| \langle 0, 0, 1, 0 \rangle \middle| \langle point_1, point_2, point_3, 1 \rangle \right\rangle \tag{2}$$

```
> Mrotxtransl:=(theta,point)->&lt;&lt;1,0,0,0&gt;|&lt;0,cos(theta),sin(theta),0&gt;|&lt;0,-
  sin(theta),cos(theta),0&gt;|&lt;point[1],point[2],point[3],1&gt;&gt;;
```

$$Mrotxtransl := (\theta, point) \mapsto \left\langle \langle 1, 0, 0, 0 \rangle \middle| \langle 0, \cos(\theta), \sin(\theta), 0 \rangle \middle| \langle 0, -\sin(\theta), \cos(\theta), 0 \rangle \middle| \langle point_1, point_2, point_3, 1 \rangle \right\rangle \tag{3}$$

```
> M01:=Mrotztransl(0,<0,0,q1(t)>).Mrotxtransl(Pi/2,<L1,0,0>);
```

$$M01 := \begin{bmatrix} 1 & 0 & 0 & L1 \\ 0 & 0 & -1 & 0 \\ 0 & 1 & 0 & q1(t) \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{4}$$

```
> M12:=Mrotztransl(q2(t),<0,0,0>).Mrotxtransl(-Pi/2,<L2,0,0>);
```

$$M12 := \begin{bmatrix} \cos(q2(t)) & 0 & -\sin(q2(t)) & \cos(q2(t))\ L2 \\ \sin(q2(t)) & 0 & \cos(q2(t)) & \sin(q2(t))\ L2 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{5}$$

```
> M02:=simplify(M01.M12);
```

(6)

$$M02 := \begin{bmatrix} \cos(q2(t)) & 0 & -\sin(q2(t)) & \cos(q2(t))\ L2 + L1 \\ 0 & 1 & 0 & 0 \\ \sin(q2(t)) & 0 & \cos(q2(t)) & \sin(q2(t))\ L2 + q1(t) \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{6}$$

```
> M23p:=Mrotztransl(q3(t),<0,0,0>).Mrotxtransl(0,<L3,0,0>);
```

$$M23p := \begin{bmatrix} \cos(q3(t)) & -\sin(q3(t)) & 0 & \cos(q3(t))\ L3 \\ \sin(q3(t)) & \cos(q3(t)) & 0 & \sin(q3(t))\ L3 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{7}$$

```
> M3p3:=<<0,0,1,0>|<0,-1,0,0>|<1,0,0,0>|<0,0,0,1>>;
```

$$M3p3 := \begin{bmatrix} 0 & 0 & 1 & 0 \\ 0 & -1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{8}$$

```
> M23:=M23p.M3p3;
```

$$M23 := \begin{bmatrix} 0 & \sin(q3(t)) & \cos(q3(t)) & \cos(q3(t))\ L3 \\ 0 & -\cos(q3(t)) & \sin(q3(t)) & \sin(q3(t))\ L3 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{9}$$

```
> M03:=simplify(M02.M23);
```

$$\begin{aligned} M03 := [&[-\sin(q2(t)),\ \cos(q2(t))\ \sin(q3(t)),\ \cos(q2(t))\ \cos(q3(t)), \\ &(\cos(q3(t))\ L3 + L2)\ \cos(q2(t)) + L1], \\ &[0,\ -\cos(q3(t)),\ \sin(q3(t)),\ \sin(q3(t))\ L3], \\ &[\cos(q2(t)),\ \sin(q2(t))\ \sin(q3(t)),\ \sin(q2(t))\ \cos(q3(t)), \\ &(\cos(q3(t))\ L3 + L2)\ \sin(q2(t)) + q1(t)], \\ &[0,\ 0,\ 0,\ 1]] \end{aligned} \tag{10}$$

```
> E_3:=<0,0,0,1>;
```

$$E\_3 := \begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix} \tag{11}$$

```
> E:=simplify(M03.E_3);
```

$$E := \begin{bmatrix} (\cos(q3(t))\ L3 + L2)\ \cos(q2(t)) + L1 \\ \sin(q3(t))\ L3 \\ (\cos(q3(t))\ L3 + L2)\ \sin(q2(t)) + q1(t) \\ 1 \end{bmatrix} \tag{12}$$

```
> Jac_E:=simplify(VectorCalculus[Jacobian](subs(q1(t)=q1,q2(t)=q2,q3(t)=
  q3,E[1..3]),[q1,q2,q3]));
```

$$Jac\_E := \begin{bmatrix} 0 & -(\cos(q3)\ L3 + L2)\ \sin(q2) & -\sin(q3)\ L3\ \cos(q2) \\ 0 & 0 & \cos(q3)\ L3 \\ 1 & (\cos(q3)\ L3 + L2)\ \cos(q2) & -\sin(q3)\ L3\ \sin(q2) \end{bmatrix} \tag{13}$$

```
> det_Jac:=Determinant(Jac_E);
```

$$det\_Jac := -(\cos(q3)\ L3 + L2)\ \sin(q2)\ \cos(q3)\ L3 \tag{14}$$

```
> q2_sglConfig:=solve(det_Jac=0,q2);
```

$$q2\_sglConfig := 0 \tag{15}$$

```
> q3_sglConfig:=solve(det_Jac=0,q3);
```

$$q3\_sglConfig := \pi - \arccos\left(\frac{L2}{L3}\right),\ \frac{\pi}{2} \tag{16}$$

```
> data:={L1=1,L2=0.6,L3=0.4,m1=15,m2=7,m3=5,K1=20000,K2=5000,K3=5000,g=
  9.81};
```

$$data := \{K1 = 20000,\ K2 = 5000,\ K3 = 5000,\ L1 = 1,\ L2 = 0.6,\ L3 = 0.4,\ g = 9.81,\ m1 = 15,\ m2 = 7,\ m3 = 5\} \tag{17}$$

```
> E
```

$$\begin{bmatrix} (\cos(q3(t))\ L3 + L2)\ \cos(q2(t)) + L1 \\ \sin(q3(t))\ L3 \\ (\cos(q3(t))\ L3 + L2)\ \sin(q2(t)) + q1(t) \\ 1 \end{bmatrix} \tag{18}$$

```
> Ex:=subs(data,q2(t)=q2,q3(t)=q3,E[1]);
```

$$Ex := (0.4\ \cos(q3) + 0.6)\ \cos(q2) + 1 \tag{19}$$

```
> Ey:=subs(data,q3(t)=q3,E[2]);
```

$$Ey := 0.4\ \sin(q3) \tag{20}$$

```
> Ez:=subs(data,q1(t)=q1,q2(t)=q2,q3(t)=q3,E[3]);
```

$$Ez := (0.4\ \cos(q3) + 0.6)\ \sin(q2) + q1 \tag{21}$$

```
> solutions:=solve({Ex=1.765,Ey=0.283,Ez=0.541},{q1,q2,q3});
```

$$solutions := \{q1 = 0.1006496793,\ q2 = 0.5223011036,\ q3 = 0.7859544135\},\ \{q1 = 0.5410000000 - 0.6960864924\ I,\ q2 = 1.527041703\ I,\ q3 = 2.355638240\} \tag{22}$$

```
> config:=solutions[1];
```

$$config := \{q1 = 0.1006496793,\ q2 = 0.5223011036,\ q3 = 0.7859544135\} \tag{23}$$

```
> Tmin_sys:=2;
```

$$Tmin\_sys := 2 \tag{24}$$

```
> amax_rescaled:=4*deltaq/(T^2);
```

$$amax\_rescaled := \frac{4\ deltaq}{T^2} \tag{25}$$

```
> amax1:=subs(deltaq=q1-0,T=Tmin_sys,config,amax_rescaled);
```

$$amax1 := 0.1006496793 \tag{26}$$

```
> amax2:=subs(deltaq=q2-0,T=Tmin_sys,config,amax_rescaled);
```

$$amax2 := 0.5223011036 \tag{27}$$

```
> amax3:=subs(deltaq=q3-0,T=Tmin_sys,config,amax_rescaled);
```

$$amax3 := 0.7859544135 \tag{28}$$

```
> base_profile:=piecewise(t>=0 and t<=T/2, qini+amax*(t^2)/2,t>T/2 and
  t<=T, qini+amax*(T^2)/4-amax*((T-t)^2)/2);
```

$$base\_profile := \begin{cases} qini + \frac{amax\ t^2}{2} & 0 \le t \le \frac{T}{2} \\ qini + \frac{amax\ T^2}{4} - \frac{amax\ (T-t)^2}{2} & \frac{T}{2} < t \le T \end{cases} \tag{29}$$

```
> q1_profile:=subs(amax=amax1,T=Tmin_sys,qini=0,base_profile);
```

$$q1\_profile := \begin{cases} 0.05032483965\ t^2 & 0 \le t \le 1 \\ 0.1006496793 - 0.05032483965\ (2-t)^2 & 1 < t \le 2 \end{cases} \tag{30}$$

```
> q2_profile:=subs(amax=amax2,T=Tmin_sys,qini=0,base_profile);
```

$$q2\_profile := \begin{cases} 0.2611505518\ t^2 & 0 \le t \le 1 \\ 0.5223011036 - 0.2611505518\ (2-t)^2 & 1 < t \le 2 \end{cases} \tag{31}$$

```
> q3_profile:=subs(amax=amax3,T=Tmin_sys,qini=0,base_profile);
```

$$q3\_profile := \begin{cases} 0.3929772068\ t^2 & 0 \le t \le 1 \\ 0.7859544136 - 0.3929772068\ (2-t)^2 & 1 < t \le 2 \end{cases} \tag{32}$$

```
> plot(q1_profile,t=0..Tmin_sys);
```




```
> plot(q2_profile,t=0..Tmin_sys);
```




```
> plot(q3_profile,t=0..Tmin_sys);
```




```
> data
```

$$\{K1 = 20000,\ K2 = 5000,\ K3 = 5000,\ L1 = 1,\ L2 = 0.6,\ L3 = 0.4,\ g = 9.81,\ m1 = 15,\ m2 = 7,\ m3 = 5\} \tag{33}$$

```
> T_func:=(W,J)->simplify(Trace(1/2*W.J.(W^%T)));
```

$$T\_func := (W,\ J) \mapsto simplify\left(LinearAlgebra:-Trace\left(\left(\frac{W}{2}\right) \cdot J \cdot W^{T}\right)\right) \tag{34}$$

```
> U_func:=(Hg,J)->simplify(Trace((-1)*Hg.J));
```

$$U\_func := (Hg,\ J) \mapsto simplify(LinearAlgebra:-Trace((-Hg) \cdot J)) \tag{35}$$

```
> W_func:=(M)->simplify(map(diff,M,t).MatrixInverse(M));
```

$$W\_func := M \mapsto simplify(map(diff,\ M,\ t) \cdot LinearAlgebra:-MatrixInverse(M)) \tag{36}$$

```
> J_template:=<<Ixx,Ixy,Ixz,m*Xg>|<Iyx,Iyy,Iyz,m*Yg>|<Izx,Izy,Izz,m*
  Zg>|<m*Xg,m*Yg,m*Zg,m>>;
```

$$J\_template := \begin{bmatrix} Ixx & Iyx & Izx & m\ Xg \\ Ixy & Iyy & Izy & m\ Yg \\ Ixz & Iyz & Izz & m\ Zg \\ m\ Xg & m\ Yg & m\ Zg & m \end{bmatrix} \tag{37}$$

```
> J_mobile_2_fixed:=(M,J)->simplify(M.J.(M^%T));
```

$$J\_mobile\_2\_fixed := (M,\ J) \mapsto simplify\left(M \cdot J \cdot M^{T}\right) \tag{38}$$

```
> I_func:=(a,b,m)->a*b*m;
```

$$I\_func := (a,\ b,\ m) \mapsto m \cdot a \cdot b \tag{39}$$

```
> Hg_template:=<<0,0,0,0>|<0,0,0,0>|<0,0,0,0>|<0,0,-g,0>>;
```

$$Hg\_template := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -g \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{40}$$

```
> W01:=W_func(M01);
```

$$W01 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{d}{dt} q1(t) \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{41}$$

```
> W02:=W_func(M02);
```

$$W02 := \begin{bmatrix} 0 & 0 & -\frac{d}{dt} q2(t) & \left(\frac{d}{dt} q2(t)\right) q1(t) \\ 0 & 0 & 0 & 0 \\ \frac{d}{dt} q2(t) & 0 & 0 & -L1 \left(\frac{d}{dt} q2(t)\right) + \frac{d}{dt} q1(t) \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{42}$$

```
> W03:=W_func(M03);
```

$$W03 := \left[\left[0,\ -\left(\frac{d}{dt} q3(t)\right) \cos(q2(t)),\ -\frac{d}{dt} q2(t),\ \left(\frac{d}{dt} q2(t)\right) q1(t)\right],\right.$$
$$\left[\left(\frac{d}{dt} q3(t)\right) \cos(q2(t)),\ 0,\ \left(\frac{d}{dt} q3(t)\right) \sin(q2(t)),\ -\left(\frac{d}{dt} q3(t)\right) (L1 \cos(q2(t)) + \sin(q2(t))\ q1(t) + L2)\right],$$
$$\left[\frac{d}{dt} q2(t),\ -\left(\frac{d}{dt} q3(t)\right) \sin(q2(t)),\ 0,\ -L1 \left(\frac{d}{dt} q2(t)\right) + \frac{d}{dt} q1(t)\right],$$
$$\left.\left[0,\ 0,\ 0,\ 0\right]\right] \tag{43}$$

```
> J1_1:=subs(Ixx=I_func(-L1/2,-L1/2,m1),Ixy=0,Ixz=0,Iyx=0,Iyy=0,Iyz=0,
  Izx=0,Izy=0,Izz=0,m=m1,Xg=-L1/2,Yg=0,Zg=0,J_template);
```

$$J1\_1 := \begin{bmatrix} \frac{m1\ L1^2}{4} & 0 & 0 & -\frac{m1\ L1}{2} \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ -\frac{m1\ L1}{2} & 0 & 0 & m1 \end{bmatrix} \tag{44}$$

```
> J2_2:=subs(Ixx=I_func(-L2/2,-L2/2,m2),Ixy=0,Ixz=0,Iyx=0,Iyy=0,Iyz=0,
  Izx=0,Izy=0,Izz=0,m=m2,Xg=-L2/2,Yg=0,Zg=0,J_template);
```

$$J2\_2 := \begin{bmatrix} \frac{m2\ L2^2}{4} & 0 & 0 & -\frac{m2\ L2}{2} \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ -\frac{m2\ L2}{2} & 0 & 0 & m2 \end{bmatrix} \tag{45}$$

```
> J3_3:=subs(Ixx=0,Ixy=0,Ixz=0,Iyx=0,Iyy=0,Iyz=0,Izx=0,Izy=0,Izz=0,m=m3,
  Xg=0,Yg=0,Zg=0,J_template);
```

$$J3\_3 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & m3 \end{bmatrix} \tag{46}$$

```
> J1:=J_mobile_2_fixed(M01,J1_1);
```

$$J1 := \begin{bmatrix} \frac{m1\ L1^2}{4} & 0 & \frac{q1(t)\ m1\ L1}{2} & \frac{m1\ L1}{2} \\ 0 & 0 & 0 & 0 \\ \frac{q1(t)\ m1\ L1}{2} & 0 & q1(t)^2\ m1 & q1(t)\ m1 \\ \frac{m1\ L1}{2} & 0 & q1(t)\ m1 & m1 \end{bmatrix} \tag{47}$$

```
> J2:=J_mobile_2_fixed(M02,J2_2);
```

$$J2 := \left[\left[\frac{m2\ (\cos(q2(t))\ L2 + 2\ L1)^2}{4},\ 0,\ \frac{m2\ (\sin(q2(t))\ L2 + 2\ q1(t))\ (\cos(q2(t))\ L2 + 2\ L1)}{4},\ \frac{m2\ (\cos(q2(t))\ L2 + 2\ L1)}{2} \right], \left[0,\ 0,\ 0,\ 0 \right], \left[\frac{m2\ (\sin(q2(t))\ L2 + 2\ q1(t))\ (\cos(q2(t))\ L2 + 2\ L1)}{4},\ 0,\ \frac{m2\ (\sin(q2(t))\ L2 + 2\ q1(t))^2}{4},\ \frac{m2\ (\sin(q2(t))\ L2 + 2\ q1(t))}{2} \right], \right. \tag{48}$$

$$\left[\frac{m2\ (\cos(q2(t))\ L2+2\ L1)}{2},\ 0,\ \frac{m2\ (\sin(q2(t))\ L2+2\ q1(t))}{2},\ m2\right]\Big]$$

```
> J3:=J_mobile_2_fixed(M03,J3_3);
```

$$J3 := [[((\cos(q3(t))\ L3+L2)\ \cos(q2(t))+L1)^2\ m3,\ ((\cos(q3(t))\ L3+L2)\ \cos(q2(t))+L1)\ m3\ \sin(q3(t))\ L3,\ m3\ ((\cos(q3(t))\ L3+L2)\ \cos(q2(t))+L1)\ (\sin(q2(t))\ \cos(q3(t))\ L3+\sin(q2(t))\ L2+q1(t)),\ ((\cos(q3(t))\ L3+L2)\ \cos(q2(t))+L1)\ m3],$$
$$[((\cos(q3(t))\ L3+L2)\ \cos(q2(t))+L1)\ m3\ \sin(q3(t))\ L3,\ \sin(q3(t))^2\ L3^2\ m3,\ \sin(q3(t))\ L3\ m3\ ((\cos(q3(t))\ L3+L2)\ \sin(q2(t))+q1(t)),\ \sin(q3(t))\ L3\ m3],$$
$$[m3\ ((\cos(q3(t))\ L3+L2)\ \cos(q2(t))+L1)\ (\sin(q2(t))\ \cos(q3(t))\ L3+\sin(q2(t))\ L2+q1(t)),\ \sin(q3(t))\ L3\ m3\ ((\cos(q3(t))\ L3+L2)\ \sin(q2(t))+q1(t)),\ ((\cos(q3(t))\ L3+L2)\ \sin(q2(t))+q1(t))^2\ m3,\ ((\cos(q3(t))\ L3+L2)\ \sin(q2(t))+q1(t))\ m3],$$
$$[((\cos(q3(t))\ L3+L2)\ \cos(q2(t))+L1)\ m3,\ \sin(q3(t))\ L3\ m3,\ ((\cos(q3(t))\ L3+L2)\ \sin(q2(t))+q1(t))\ m3,\ m3]] \tag{49}$$

```
> T1:=T_func(W01,J1);
```

$$T1 := \frac{\left(\frac{d}{dt}\ q1(t)\right)^2\ m1}{2} \tag{50}$$

```
> T2:=T_func(W02,J2);
```

$$T2 := \frac{1}{8}\left(m2\left(4\left(\frac{d}{dt}\ q1(t)\right)^2+L2^2\left(\frac{d}{dt}\ q2(t)\right)^2+4\left(\frac{d}{dt}\ q2(t)\right)\ L2\left(\frac{d}{dt}\ q1(t)\right)\ \cos(q2(t))\right)\right) \tag{51}$$

```
> T3:=T_func(W03,J3);
```

$$T3 := -m3\left(-\frac{(\cos(q3(t))\ L3+L2)^2\left(\frac{d}{dt}\ q2(t)\right)^2}{2}-\left(\frac{d}{dt}\ q1(t)\right)\cos(q2(t))\ (\cos(q3(t))\ L3+L2)\left(\frac{d}{dt}\ q2(t)\right)+\left(\frac{d}{dt}\ q3(t)\right)\sin(q2(t))\ \sin(q3(t))\ L3\left(\frac{d}{dt}\ q1(t)\right)-\frac{\left(\frac{d}{dt}\ q3(t)\right)^2\ L3^2}{2}\right) \tag{52}$$

$$-\frac{\left(\frac{d}{dt}\,q1(t)\right)^2}{2}\Bigg)$$

```
> Hg1:=Hg_template;
```

$$Hg1 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -g \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{53}$$

```
> Hg2:=Hg_template;
```

$$Hg2 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -g \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{54}$$

```
> Hg3:=Hg_template;
```

$$Hg3 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -g \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{55}$$

```
> U1:=U_func(Hg1,J1);
```

$$U1 := g\ q1(t)\ m1 \tag{56}$$

```
> U2:=U_func(Hg2,J2);
```

$$U2 := \frac{g\ m2\ (\sin(q2(t))\ L2 + 2\ q1(t))}{2} \tag{57}$$

```
> U3:=U_func(Hg3,J3);
```

$$U3 := g\ ((\cos(q3(t))\ L3 + L2)\ \sin(q2(t)) + q1(t))\ m3 \tag{58}$$

```
> Lagr:=T1+T2+T3-U1-U2-U3;
```

$$Lagr := \frac{\left(\frac{d}{dt}\,q1(t)\right)^2\ m1}{2} + \frac{1}{8}\Bigg(m2\left(4\left(\frac{d}{dt}\,q1(t)\right)^2 + L2^2\left(\frac{d}{dt}\,q2(t)\right)^2 + 4\left(\frac{d}{dt}\,q2(t)\right)\ L2\left(\frac{d}{dt}\,q1(t)\right)\cos(q2(t))\right)\Bigg) - m3\Bigg(-\frac{(\cos(q3(t))\ L3 + L2)^2\left(\frac{d}{dt}\,q2(t)\right)^2}{2} - \left(\frac{d}{dt}\right. \tag{59}$$

$$q1(t)\Big)\ \cos(q2(t))\ (\cos(q3(t))\ L3+L2)\ \left(\frac{d}{dt}\ q2(t)\right)+\left(\frac{d}{dt}$$

$$q3(t)\Big)\ \sin(q2(t))\ \sin(q3(t))\ L3\ \left(\frac{d}{dt}\ q1(t)\right)-\frac{\left(\frac{d}{dt}\ q3(t)\right)^2\ L3^2}{2}$$

$$-\frac{\left(\frac{d}{dt}\ q1(t)\right)^2}{2}\Bigg)-g\ q1(t)\ m1-\frac{g\ m2\ (\sin(q2(t))\ L2+2\ q1(t))}{2}$$

$$-g\ ((\cos(q3(t))\ L3+L2)\ \sin(q2(t))+q1(t))\ m3$$

```
> diffF := (f,x)->subs(y=x,diff(subs(x=y,f),y));
```

$$diffF := (f,\ x)\rightarrow subs\left(y=x,\ \frac{\partial}{\partial y}\ subs(x=y,\ f)\right) \tag{60}$$

```
> EQ_left:=(Lagr,qt)->simplify(diff(diffF(Lagr,diff(qt,t)),t)-diffF(Lagr,
  qt));
```

$$EQ\_left := (Lagr,\ qt)\rightarrow simplify\left(\frac{\partial}{\partial t}\ diffF\left(Lagr,\ \frac{\partial}{\partial t}\ qt\right)-diffF(Lagr,\ qt)\right) \tag{61}$$

```
> EQ_1_left:=EQ_left(Lagr,q1(t));
```

$$EQ\_1\_left := \frac{(2\ m1+2\ m2+2\ m3)\ \left(\frac{d^2}{dt^2}\ q1(t)\right)}{2}+\Bigg(\cos(q3(t))\ L3\ m3 \tag{62}$$

$$+\frac{L2\ (m2+2\ m3)}{2}\Bigg)\ \cos(q2(t))\ \left(\frac{d^2}{dt^2}\ q2(t)\right)-L3\ \left(\frac{d^2}{dt^2}\right.$$

$$q3(t)\Big)\ \sin(q3(t))\ \sin(q2(t))\ m3-\sin(q2(t))\ \Bigg(\cos(q3(t))\ L3\ m3$$

$$+\frac{L2\ (m2+2\ m3)}{2}\Bigg)\ \left(\frac{d}{dt}\ q2(t)\right)^2-2\ L3\ \left(\frac{d}{dt}\ q3(t)\right)\ \sin(q3(t))\ \left(\frac{d}{dt}\right.$$

$$q2(t)\Big)\ \cos(q2(t))\ m3-L3\ \left(\frac{d}{dt}\ q3(t)\right)^2\ \cos(q3(t))\ \sin(q2(t))\ m3+g\ (m1$$

$$+m2+m3)$$

```
> EQ_2_left:=EQ_left(Lagr,q2(t));
```

$$EQ\_2\_left := \tag{63}$$

$$\frac{1}{4}\left(\left(4\ \cos(q3(t))^2\ L3^2\ m3+8\ \cos(q3(t))\ L2\ L3\ m3+L2^2\ (m2\right.\right.$$

$$\left.\left.+4\ m3)\right)\ \left(\frac{d^2}{dt^2}\ q2(t)\right)\right)+\Bigg(\cos(q3(t))\ L3\ m3$$

$$+\frac{L2\ (m2+2\ m3)}{2}\Bigg)\ \cos(q2(t))\ \left(\frac{d^2}{dt^2}\ q1(t)\right)-2\ \sin(q3(t))\ \left(\frac{d}{dt}\ q3(t)\right)\ L3\ m3\ (\cos(q3(t))\ L3+L2)\ \left(\frac{d}{dt}\ q2(t)\right)+\left(\cos(q3(t))\ L3\ m3+\frac{L2\ (m2+2\ m3)}{2}\right)\ g\ \cos(q2(t))$$

```
> EQ_3_left:=EQ_left(Lagr,q3(t));
```

$$EQ\_3\_left := -m3\left(\sin(q2(t))\ \sin(q3(t))\ \left(\frac{d^2}{dt^2}\ q1(t)\right)-\left(\frac{d^2}{dt^2}\ q3(t)\right)\ L3+\left(\left(-\cos(q3(t))\ L3-L2\right)\ \left(\frac{d}{dt}\ q2(t)\right)^2+g\ \sin(q2(t))\right)\ \sin(q3(t))\right)\ L3 \tag{64}$$

```
> EQ1:=EQ_1_left-F1;
```

$$EQ1 := \frac{(2\ m1+2\ m2+2\ m3)\ \left(\frac{d^2}{dt^2}\ q1(t)\right)}{2}+\left(\cos(q3(t))\ L3\ m3+\frac{L2\ (m2+2\ m3)}{2}\right)\ \cos(q2(t))\ \left(\frac{d^2}{dt^2}\ q2(t)\right)-L3\ \left(\frac{d^2}{dt^2}\ q3(t)\right)\ \sin(q3(t))\ \sin(q2(t))\ m3-\sin(q2(t))\ \left(\cos(q3(t))\ L3\ m3+\frac{L2\ (m2+2\ m3)}{2}\right)\ \left(\frac{d}{dt}\ q2(t)\right)^2-2\ L3\ \left(\frac{d}{dt}\ q3(t)\right)\ \sin(q3(t))\ \left(\frac{d}{dt}\ q2(t)\right)\ \cos(q2(t))\ m3-L3\ \left(\frac{d}{dt}\ q3(t)\right)^2\ \cos(q3(t))\ \sin(q2(t))\ m3+g\ (m1+m2+m3)-F1 \tag{65}$$

```
> EQ2:=EQ_2_left-C2;
```

$$EQ2 := \frac{1}{4}\left(\left(4\ \cos(q3(t))^2\ L3^2\ m3+8\ \cos(q3(t))\ L2\ L3\ m3+L2^2\ (m2+4\ m3)\right)\ \left(\frac{d^2}{dt^2}\ q2(t)\right)\right)+\left(\cos(q3(t))\ L3\ m3+\frac{L2\ (m2+2\ m3)}{2}\right)\ \cos(q2(t))\ \left(\frac{d^2}{dt^2}\ q1(t)\right)-2\ \sin(q3(t))\ \left(\frac{d}{dt}\ q3(t)\right)\ L3\ m3\ (\cos(q3(t))\ L3+L2)\ \left(\frac{d}{dt}\ q2(t)\right)+\left(\cos(q3(t))\ L3\ m3\right. \tag{66}$$

$$+\frac{L2\ (m2+2\ m3)}{2}\Bigg)\ g\ \cos(q2(t))-C2$$

```
> EQ3:=EQ_3_left-C3;
```

$$EQ3 := -m3\left(\sin(q2(t))\ \sin(q3(t))\left(\frac{d^2}{dt^2}\ q1(t)\right)-\left(\frac{d^2}{dt^2}\ q3(t)\right)\ L3+\Bigg(\Big(-\cos(q3(t))\ L3-L2\Big)\left(\frac{d}{dt}\ q2(t)\right)^2+g\ \sin(q2(t))\Bigg)\ \sin(q3(t))\right)\ L3-C3 \tag{67}$$

```
> f1:=simplify(solve(subs(data,EQ1)=0,F1));
```

$$f1 := (2.\ \cos(q3(t))+5.100000000)\ \cos(q2(t))\left(\frac{d^2}{dt^2}\ q2(t)\right)+27.\ \frac{d^2}{dt^2}\ q1(t)-2.\ \left(\frac{d^2}{dt^2}\ q3(t)\right)\ \sin(q3(t))\ \sin(q2(t))+(-2.\ \cos(q3(t))-5.100000000)\ \sin(q2(t))\left(\frac{d}{dt}\ q2(t)\right)^2-4.\ \left(\frac{d}{dt}\ q3(t)\right)\ \sin(q3(t))\left(\frac{d}{dt}\ q2(t)\right)\cos(q2(t))+264.8700000-2.\ \left(\frac{d}{dt}\ q3(t)\right)^2\cos(q3(t))\ \sin(q2(t)) \tag{68}$$

```
> c2:=simplify(solve(subs(data,EQ2)=0,C2));
```

$$c2 := \left(0.8000000000\ \cos(q3(t))^2+2.400000000\ \cos(q3(t))+2.430000000\right)\left(\frac{d^2}{dt^2}\ q2(t)\right)+(2.\ \cos(q2(t))\ \cos(q3(t))+5.100000000\ \cos(q2(t)))\left(\frac{d^2}{dt^2}\ q1(t)\right)+(-1.600000000\ \sin(q3(t))\ \cos(q3(t))-2.400000000\ \sin(q3(t)))\left(\frac{d}{dt}\ q3(t)\right)\left(\frac{d}{dt}\ q2(t)\right)+19.62000000\ \cos(q2(t))\ \cos(q3(t))+50.03100000\ \cos(q2(t)) \tag{69}$$

```
> c3:=simplify(solve(subs(data,EQ3)=0,C3));
```

$$c3 := -2.\ \sin(q2(t))\ \sin(q3(t))\left(\frac{d^2}{dt^2}\ q1(t)\right)+0.8000000000\ \frac{d^2}{dt^2}\ q3(t)+(0.8000000000\ \cos(q3(t))+1.200000000)\ \sin(q3(t))\left(\frac{d}{dt}\ q2(t)\right)^2-19.62000000\ \sin(q2(t))\ \sin(q3(t)) \tag{70}$$

```
> f1_time:=simplify(subs(q1(t)=q1_profile,q2(t)=q2_profile,q3(t)=
  q3_profile,f1));
```

$$f1\_time := \left\{ \left(-1.044602207 \cos\left(0.5223011034 + 0.2611505518\ t^2 - 1.044602207\ t\right) + \left(1.78104556\right.\right.\right.$$

```
> c2_time:=simplify(subs(q1(t)=q1_profile,q2(t)=q2_profile,q3(t)=
  q3_profile,c2));
```

$$c2\_time := \left\{ \left(\left(0.6568077716\ t^2 - 2.627231088\ t + 2.627231090\right) \cos\left(0.7859544134 + 0.3929772068\right.\right.\right.$$

```
> c3_time:=simplify(subs(q1(t)=q1_profile,q2(t)=q2_profile,q3(t)=
  q3_profile,c3));
```

$$c3\_time := \begin{cases} 0. & \\ -0.6287635302 + \left(-19.41870064\ \sin\left(0.5223011034 + 0.2611505518\ t^2 - 1.044602207\ t\right.\right. & \end{cases}$$

```
> plot(f1_time,t=0..Tmin_sys);
```




```
> plot(c2_time,t=0..Tmin_sys);
```




```
> plot(c3_time,t=0..Tmin_sys);
```




```
> t1:=0.1*10^(-3);
```

$$t1 := 0.0001000000000 \tag{74}$$

```
> t2:=999.9*10^(-3);
```

$$t2 := 0.9999000000 \tag{75}$$

```
> t3:=1000.1*10^(-3);
```

$$t3 := 1.000100000 \tag{76}$$

```
> t4:=1999.9*10^(-3);
```

$$t4 := 1.999900000 \tag{77}$$

```
> evalf(subs(t=t1,f1_time));
```

$$271.2958791 \tag{78}$$

```
> evalf(subs(t=t2,f1_time));
```

$$269.5469708 \tag{79}$$

```
> evalf(subs(t=t3,f1_time));
```

$$257.4108839 \tag{80}$$

```
> evalf(subs(t=t4,f1_time));
```

$$259.7588739 \tag{81}$$

```
> evalf(subs(t=t1,c2_time));
```

$$73.30616793 \tag{82}$$

```
> evalf(subs(t=t2,c2_time));
```

$$68.69597966 \tag{83}$$

```
> evalf(subs(t=t3,c2_time));
```

$$61.77420862 \tag{84}$$

```
> evalf(subs(t=t4,c2_time));
```

$$52.44566133 \tag{85}$$

```
> evalf(subs(t=t1,c3_time));
```

$$0.6287635302 \tag{86}$$

```
> evalf(subs(t=t2,c3_time));
```

$$-1.127774080 \tag{87}$$

```
> evalf(subs(t=t3,c3_time));
```

$$-2.346917280 \tag{88}$$

```
> evalf(subs(t=t4,c3_time));
```

$$-7.482683226 \tag{89}$$

```
> data
```

$$\{K1 = 20000,\ K2 = 5000,\ K3 = 5000,\ L1 = 1,\ L2 = 0.6,\ L3 = 0.4,\ g = 9.81,\ m1 = 15,\ m2 = 7,\ m3 = 5\} \tag{90}$$

```
> EQ1_controlled:=subs(F1=K1*(q1_profile-q1(t)),data,EQ1);
```

$$EQ1\_controlled := 27 \left(\frac{d^2}{dt^2} q1(t)\right) + (2.0 \cos(q3(t)) + 5.100000000) \cos(q2(t)) \left(\frac{d^2}{dt^2} q2(t)\right) - 2.0 \left(\frac{d^2}{dt^2} q3(t)\right) \sin(q3(t)) \sin(q2(t)) - \sin(q2(t)) (2.0 \cos(q3(t)) + 5.100000000) \left(\frac{d}{dt} q2(t)\right)^2 - 4.0 \left(\frac{d}{dt} q3(t)\right) \sin(q3(t)) \left(\frac{d}{dt} q2(t)\right) \cos(q2(t)) - 2.0 \left(\frac{d}{dt} q3(t)\right)^2 \cos(q3(t)) \sin(q2(t)) + 264.87 \tag{91}$$

$$-20000\left(\begin{cases} 0.05032483965\ t^2 & 0 \le t \le 1 \\ 0.1006496793 - 0.05032483965\ (2-t)^2 & 1 < t \le 2 \end{cases}\right) + 20000\ q1(t)$$

```
> EQ2_controlled:=subs(C2=K2*(q2_profile-q2(t)),data,EQ2);
```

$$EQ2\_controlled := \frac{\left(3.20\ \cos(q3(t))^2 + 9.60\ \cos(q3(t)) + 9.72\right)\left(\frac{d^2}{dt^2}\ q2(t)\right)}{4} \tag{92}$$

$$+ (2.0\ \cos(q3(t)) + 5.100000000)\ \cos(q2(t))\left(\frac{d^2}{dt^2}\ q1(t)\right)$$

$$-4.0\ \sin(q3(t))\left(\frac{d}{dt}\ q3(t)\right)(0.4\ \cos(q3(t)) + 0.6)\left(\frac{d}{dt}\ q2(t)\right)$$

$$+9.81\ (2.0\ \cos(q3(t)) + 5.100000000)\ \cos(q2(t))$$

$$-5000\left(\begin{cases} 0.2611505518\ t^2 & 0 \le t \le 1 \\ 0.5223011036 - 0.2611505518\ (2-t)^2 & 1 < t \le 2 \end{cases}\right) + 5000\ q2(t)$$

```
> EQ3_controlled:=subs(C3=K3*(q3_profile-q3(t)),data,EQ3);
```

$$EQ3\_controlled := -2.0\ \sin(q2(t))\ \sin(q3(t))\left(\frac{d^2}{dt^2}\ q1(t)\right) + 0.80\ \frac{d^2}{dt^2}\ q3(t) \tag{93}$$

$$-2.0\left((-0.4\ \cos(q3(t)) - 0.6)\left(\frac{d}{dt}\ q2(t)\right)^2 + 9.81\ \sin(q2(t))\right)\sin(q3(t))$$

$$-5000\left(\begin{cases} 0.3929772068\ t^2 & 0 \le t \le 1 \\ 0.7859544136 - 0.3929772068\ (2-t)^2 & 1 < t \le 2 \end{cases}\right) + 5000\ q3(t)$$

```
> Q_controlled:=dsolve({EQ1_controlled,EQ2_controlled,EQ3_controlled,q1
  (0)=0,q2(0)=0,q3(0)=0,D(q1)(0)=0,D(q2)(0)=0,D(q3)(0)=0},{q1(t),q2(t),q3
  (t)},numeric,output=listprocedure);
```

$$Q\_controlled := \left[t = \text{proc}(t)\ \ldots\ \text{end proc},\ q1(t) = \text{proc}(t)\ \ldots\ \text{end proc},\ \frac{d}{dt}\right. \tag{94}$$

$$q1(t) = \text{proc}(t)\ \ldots\ \text{end proc},\ q2(t) = \text{proc}(t)\ \ldots\ \text{end proc},\ \frac{d}{dt}\ q2(t) =$$

$$\text{proc}(t)$$

$$\ldots$$

$$\text{end proc},\ q3(t) = \text{proc}(t)\ \ldots\ \text{end proc},\ \frac{d}{dt}\,q3(t) = \text{proc}(t)\ \ldots\ \text{end proc}$$

```
> q1_controlled:=rhs(Q_controlled[2]);
```

$$q1\_controlled := \text{proc}(t)\ \ldots\ \text{end proc} \tag{95}$$

```
> q2_controlled:=rhs(Q_controlled[4]);
```

$$q2\_controlled := \text{proc}(t)\ \ldots\ \text{end proc} \tag{96}$$

```
> q3_controlled:=rhs(Q_controlled[6]);
```

$$q3\_controlled := \text{proc}(t)\ \ldots\ \text{end proc} \tag{97}$$

```
> plot([q1_controlled(t),q1_profile],t=0..Tmin_sys);
```




```
> plot([q2_controlled(t),q2_profile],t=0..Tmin_sys);
```




```
> plot([q3_controlled(t),q3_profile],t=0..Tmin_sys);
```




```
> q1val_0s:=evalf(q1_controlled(0));
```

$$q1val\_0s := 0. \tag{98}$$

```
> q2val_0s:=evalf(q2_controlled(0));
```

$$q2val\_0s := 0. \tag{99}$$

```
> q3val_0s:=evalf(q3_controlled(0));
```

$$q3val\_0s := 0. \tag{100}$$

```
> q1val_1s:=evalf(q1_controlled(1));
```

$$q1val\_1s := 0.0296921803302205 \tag{101}$$

```
> q2val_1s:=evalf(q2_controlled(1));
```

$$q2val\_1s := 0.227478316362466 \tag{102}$$

```
> q3val_1s:=evalf(q3_controlled(1));
```

$$q3val\_1s := 0.392993471506413 \tag{103}$$

```
> q1val_2s:=evalf(q1_controlled(2));
```

$$q1val\_2s := 0.0810687681208569 \tag{104}$$

```
> q2val_2s:=evalf(q2_controlled(2));
```

$$q2val\_2s := 0.495169112945155 \tag{105}$$

```
> q3val_2s:=evalf(q3_controlled(2));
```

$$q3val\_2s := 0.787340814164724 \tag{106}$$