RPP Robot Modeling




```
> restart:with(LinearAlgebra):
```

Position Analysis

```
> Mrotztransl:=(alpha,point)->&lt;&lt;cos(alpha),sin(alpha),0,0&gt;|&lt;-sin(alpha),
  cos(alpha),0,0&gt;|&lt;0,0,1,0&gt;|&lt;point[1],point[2],point[3],1&gt;&gt;;
```

$$Mrotztransl := (\alpha, point) \mapsto \left\langle \langle \cos(\alpha), \sin(\alpha), 0, 0 \rangle \middle| \langle -\sin(\alpha), \cos(\alpha), 0, 0 \rangle \middle| \langle 0, 0, 1, 0 \rangle \middle| \langle point_1, point_2, point_3, 1 \rangle \right\rangle \quad (1)$$

```
> Mrotxtransl:=(alpha,point)->&lt;&lt;1,0,0,0&gt;|&lt;0,cos(alpha),-sin(alpha),0&gt;|&lt;0,
  sin(alpha),cos(alpha),0&gt;|&lt;point[1],point[2],point[3],1&gt;&gt;;
```

$$Mrotxtransl := (\alpha, point) \mapsto \left\langle \langle 1, 0, 0, 0 \rangle \middle| \langle 0, \cos(\alpha), -\sin(\alpha), 0 \rangle \middle| \langle 0, \sin(\alpha), \cos(\alpha), 0 \rangle \middle| \langle point_1, point_2, point_3, 1 \rangle \right\rangle \quad (2)$$

```
> M01:=simplify(Mrotztransl(q1(t),&lt;0,0,0&gt;).Mrotxtransl(Pi/2,&lt;0,0,0&gt;));
```

$$M01 := \begin{bmatrix} \cos(q1(t)) & 0 & -\sin(q1(t)) & 0 \\ \sin(q1(t)) & 0 & \cos(q1(t)) & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad (3)$$

```
> M12:=simplify(Mrotztransl(0,&lt;0,0,q2(t)&gt;).Mrotxtransl(-Pi/2,&lt;0,0,0&gt;));
```

(4)

$$M12 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 1 & 0 & q2(t) \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{4}$$

```
> M02:=simplify(M01.M12);
```

$$M02 := \begin{bmatrix} \cos(q1(t)) & -\sin(q1(t)) & 0 & -\sin(q1(t))\ q2(t) \\ \sin(q1(t)) & \cos(q1(t)) & 0 & \cos(q1(t))\ q2(t) \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{5}$$

```
> M23:=Mrotztransl(0,<0,0,q3(t)>);
```

$$M23 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & q3(t) \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{6}$$

```
> M03:=simplify(M02.M23);
```

$$M03 := \begin{bmatrix} \cos(q1(t)) & -\sin(q1(t)) & 0 & -\sin(q1(t))\ q2(t) \\ \sin(q1(t)) & \cos(q1(t)) & 0 & \cos(q1(t))\ q2(t) \\ 0 & 0 & 1 & q3(t) \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{7}$$

```
> E_3:=<0,0,0,1>;
```

$$E\_3 := \begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix} \tag{8}$$

```
> E:=M03.E_3;
```

$$E := \begin{bmatrix} -\sin(q1(t))\ q2(t) \\ \cos(q1(t))\ q2(t) \\ q3(t) \\ 1 \end{bmatrix} \tag{9}$$

```
> Jac_E:=simplify(VectorCalculus[Jacobian](subs(q1(t)=q1,q2(t)=q2,q3(t)=
  q3,E[1..3]),[q1,q2,q3]));
```

(10)

$$Jac\_E := \begin{bmatrix} -\cos(q1) & q2 & -\sin(q1) & 0 \\ -\sin(q1) & q2 & \cos(q1) & 0 \\ 0 & & 0 & 1 \end{bmatrix} \tag{10}$$

```
> W_template:=(M)->simplify(map(diff,M,t).MatrixInverse(M));
```

$$W\_template := M \mapsto simplify(map(diff,\ M,\ t) \cdot LinearAlgebra:-MatrixInverse(M)) \tag{11}$$

```
> W01:=W_template(M01);
```

$$W01 := \begin{bmatrix} 0 & -\frac{d}{dt} q1(t) & 0 & 0 \\ \frac{d}{dt} q1(t) & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{12}$$

```
> W02:=W_template(M02);
```

$$W02 := \begin{bmatrix} 0 & -\frac{d}{dt} q1(t) & 0 & -\sin(q1(t)) \left(\frac{d}{dt} q2(t)\right) \\ \frac{d}{dt} q1(t) & 0 & 0 & \cos(q1(t)) \left(\frac{d}{dt} q2(t)\right) \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{13}$$

```
> W03:=W_template(M03);
```

$$W03 := \begin{bmatrix} 0 & -\frac{d}{dt} q1(t) & 0 & -\sin(q1(t)) \left(\frac{d}{dt} q2(t)\right) \\ \frac{d}{dt} q1(t) & 0 & 0 & \cos(q1(t)) \left(\frac{d}{dt} q2(t)\right) \\ 0 & 0 & 0 & \frac{d}{dt} q3(t) \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{14}$$

```
> G1:=M01.<0,0,0.5,1>;
```

$$G1 := \begin{bmatrix} -0.5\ \sin(q1(t)) \\ 0.5\ \cos(q1(t)) \\ 0. \\ 1. \end{bmatrix} \tag{15}$$

```
> G2:=M02.<0,0,0,1>;
```

$$G2 := \begin{bmatrix} -\sin(q1(t))\ q2(t) \\ \cos(q1(t))\ q2(t) \\ 0 \\ 1 \end{bmatrix} \tag{16}$$

```
> G3:=M03.<0,0,-0.25,1>;
```

$$G3 := \begin{bmatrix} -\sin(q1(t))\ q2(t) \\ \cos(q1(t))\ q2(t) \\ -0.25 + q3(t) \\ 1. \end{bmatrix} \tag{17}$$

```
> JacG1:=simplify(VectorCalculus[Jacobian](subs(q1(t)=q1,q2(t)=q2,q3(t)=
  q3,G1[1..3]),[q1,q2,q3]));
```

$$JacG1 := \begin{bmatrix} -0.5\ \cos(q1) & 0 & 0 \\ -0.5\ \sin(q1) & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \tag{18}$$

```
> JacG2:=simplify(VectorCalculus[Jacobian](subs(q1(t)=q1,q2(t)=q2,q3(t)=
  q3,G2[1..3]),[q1,q2,q3]));
```

$$JacG2 := \begin{bmatrix} -\cos(q1)\ q2 & -\sin(q1) & 0 \\ -\sin(q1)\ q2 & \cos(q1) & 0 \\ 0 & 0 & 0 \end{bmatrix} \tag{19}$$

```
> JacG3:=simplify(VectorCalculus[Jacobian](subs(q1(t)=q1,q2(t)=q2,q3(t)=
  q3,G3[1..3]),[q1,q2,q3]));
```

$$JacG3 := \begin{bmatrix} -\cos(q1)\ q2 & -\sin(q1) & 0 \\ -\sin(q1)\ q2 & \cos(q1) & 0 \\ 0 & 0 & 1 \end{bmatrix} \tag{20}$$

```
> FqG1:=-(JacG1^%T).<0,0,m1*g>;
```

$$FqG1 := \begin{bmatrix} 0. \\ -0. \\ -0. \end{bmatrix} \tag{21}$$

```
> FqG2:=-(JacG2^%T).<0,0,m2*g>;
```

$$FqG2 := \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix} \tag{22}$$

```
> FqG3:=-(JacG3^%T).<0,0,m3*g>;
```

$$FqG3 := \begin{bmatrix} 0 \\ 0 \\ -m3\ g \end{bmatrix} \tag{23}$$

```
> J_template:=<<Ixx,Ixy,Ixz,m*Xg>|<Iyx,Iyy,Iyz,m*Yg>|<Izx,Izy,Izz,m*
  Zg>|<m*Xg,m*Yg,m*Zg,m>>;
```

$$J\_template := \begin{bmatrix} Ixx & Iyx & Izx & m\ Xg \\ Ixy & Iyy & Izy & m\ Yg \\ Ixz & Iyz & Izz & m\ Zg \\ m\ Xg & m\ Yg & m\ Zg & m \end{bmatrix} \tag{24}$$

```
> J_MobileToFixed:=(M,J)->simplify(M.J.(M^%T));
```

$$J\_MobileToFixed := (M,\ J) \mapsto simplify\left(M \cdot J \cdot M^{T}\right) \tag{25}$$

```
> T_template:=(W,J)->simplify(Trace(1/2*W.J.(W^%T)));
```

$$T\_template := (W,\ J) \mapsto simplify\left(LinearAlgebra:-Trace\left(\left(\frac{W}{2}\right) \cdot J \cdot W^{T}\right)\right) \tag{26}$$

```
> J1_1:=subs(Ixx=0,Ixy=0,Ixz=0,Iyx=0,Iyy=0,Iyz=0,Izx=0,Izy=0,Izz=(0.5^2)*
  m1,m=m1,Xg=0,Yg=0,Zg=0.5,J_template);
```

$$J1\_1 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0.25\ m1 & 0.5\ m1 \\ 0 & 0 & 0.5\ m1 & m1 \end{bmatrix} \tag{27}$$

```
> J2_2:=subs(Ixx=0,Ixy=0,Ixz=0,Iyx=0,Iyy=0,Iyz=0,Izx=0,Izy=0,Izz=0,m=m2,
  Xg=0,Yg=0,Zg=0,J_template);
```

$$J2\_2 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & m2 \end{bmatrix} \tag{28}$$

```
> J3_3:=subs(Ixx=0,Ixy=0,Ixz=0,Iyx=0,Iyy=0,Iyz=0,Izx=0,Izy=0,Izz=((-0.25)
  ^2)*m3,m=m3,Xg=0,Yg=0,Zg=-0.25,J_template);
```

$$J3\_3 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0.0625\ m3 & -0.25\ m3 \\ 0 & 0 & -0.25\ m3 & m3 \end{bmatrix} \tag{29}$$

```
> J1:=J_MobileToFixed(M01,J1_1);
```

$$J1 := \left[\left[0.25\ \sin(q1(t))^{2}\ m1,\ -0.25\ \sin(q1(t))\ m1\ \cos(q1(t)),\ 0.,\right.\right. \\ \left.\left.-0.5\ \sin(q1(t))\ m1\right], \tag{30}$$

$$\left[-0.25\ \sin(q1(t))\ m1\ \cos(q1(t)),\ 0.25\ \cos(q1(t))^2\ m1,\ 0.,\ 0.5\ \cos(q1(t))\ m1\right],$$
$$\left[0.,\ 0.,\ 0.,\ 0.\right],$$
$$\left[-0.5\ \sin(q1(t))\ m1,\ 0.5\ \cos(q1(t))\ m1,\ 0.,\ m1\right]]$$

```
> J2:=J_MobileToFixed(M02,J2_2);
```

$$J2 := [[\sin(q1(t))^2\ q2(t)^2\ m2,\ -\sin(q1(t))\ q2(t)^2\ m2\ \cos(q1(t)),\ 0,\ -\sin(q1(t))\ q2(t)\ m2],$$
$$\left[-\sin(q1(t))\ q2(t)^2\ m2\ \cos(q1(t)),\ \cos(q1(t))^2\ q2(t)^2\ m2,\ 0,\ \cos(q1(t))\ q2(t)\ m2\right],$$
$$\left[0,\ 0,\ 0,\ 0\right],$$
$$\left[-\sin(q1(t))\ q2(t)\ m2,\ \cos(q1(t))\ q2(t)\ m2,\ 0,\ m2\right]] \tag{31}$$

```
> J3:=J_MobileToFixed(M03,J3_3);
```

$$J3 := [[\sin(q1(t))^2\ q2(t)^2\ m3,\ -\sin(q1(t))\ q2(t)^2\ m3\ \cos(q1(t)),\ \sin(q1(t))\ q2(t)\ m3\ (0.25 - q3(t)),\ -\sin(q1(t))\ q2(t)\ m3],$$
$$\left[-\sin(q1(t))\ q2(t)^2\ m3\ \cos(q1(t)),\ \cos(q1(t))^2\ q2(t)^2\ m3,\ \cos(q1(t))\ q2(t)\ m3\ (-0.25 + q3(t)),\ \cos(q1(t))\ q2(t)\ m3\right],$$
$$\left[\sin(q1(t))\ q2(t)\ m3\ (0.25 - q3(t)),\ \cos(q1(t))\ q2(t)\ m3\ (-0.25 + q3(t)),\ m3\ (q3(t) - 0.2500000000)^2,\ m3\ (-0.25 + q3(t))\right],$$
$$\left[-\sin(q1(t))\ q2(t)\ m3,\ \cos(q1(t))\ q2(t)\ m3,\ m3\ (-0.25 + q3(t)),\ m3\right]] \tag{32}$$

```
> T1:=T_template(W01,J1);
```

$$T1 := 0.1250000000\ \left(\frac{d}{dt}\ q1(t)\right)^2\ m1 \tag{33}$$

```
> T2:=T_template(W02,J2);
```

$$T2 := \frac{m2\ \left(\left(\frac{d}{dt}\ q1(t)\right)^2\ q2(t)^2 + \left(\frac{d}{dt}\ q2(t)\right)^2\right)}{2} \tag{34}$$

```
> T3:=T_template(W03,J3);
```

$$T3 := \frac{m3\ \left(\left(\frac{d}{dt}\ q3(t)\right)^2 + \left(\frac{d}{dt}\ q1(t)\right)^2\ q2(t)^2 + \left(\frac{d}{dt}\ q2(t)\right)^2\right)}{2} \tag{35}$$

```
> T1_val:=evalf(subs(m1=10,diff(q1(t),t)=1,T1));
```

$$T1\_val := 1.250000000 \tag{36}$$

```
> T2_val:=evalf(subs(m2=5,diff(q1(t),t)=1,diff(q2(t),t)=1,q2(t)=1,T2));
```

$$T2\_val := 5. \tag{37}$$

```
> T3_val:=evalf(subs(m3=1,diff(q1(t),t)=1,diff(q2(t),t)=1,diff(q3(t),t)=
```

```
1,q2(t)=1,T3));
```

$$T3\_val := 1.500000000 \tag{38}$$

```
> T_totVal:=T1_val+T2_val+T3_val;
```

$$T\_totVal := 7.750000000 \tag{39}$$

Motion Planning

```
> Tmin:=sqrt(4*deltaq/amax);
```

$$Tmin := 2\sqrt{\frac{deltaq}{amax}} \tag{40}$$

```
> a_rescaled:=4*deltaq/(T^2)
```

$$a\_rescaled := \frac{4\ deltaq}{T^2} \tag{41}$$

```
> Tmin_1:=evalf(subs(amax=3,deltaq=Pi/2,Tmin));
```

$$Tmin\_1 := 1.447202509 \tag{42}$$

```
> Tmin_2:=evalf(subs(amax=4,deltaq=2-1,Tmin));
```

$$Tmin\_2 := 1.000000000 \tag{43}$$

```
> Tmin_3:=evalf(subs(amax=4,deltaq=2-1,Tmin));
```

$$Tmin\_3 := 1.000000000 \tag{44}$$

```
> Tmin_sys:=Tmin_1;
```

$$Tmin\_sys := 1.447202509 \tag{45}$$

```
> amax2_res:=evalf(subs(deltaq=2-1,T=Tmin_sys,a_rescaled));
```

$$amax2\_res := 1.909859318 \tag{46}$$

```
> amax3_res:=evalf(subs(deltaq=2-1,T=Tmin_sys,a_rescaled));
```

$$amax3\_res := 1.909859318 \tag{47}$$

```
> base_profile:=piecewise(t>=0 and t<=T/2,q_ini+amax*(t^2)/2,t>T/2 and
  t<=T,q_ini+amax*(T^2)/4-amax*((T-t)^2)/2);
```

$$base\_profile := \begin{cases} q\_ini + \frac{amax\ t^2}{2} & 0 \le t \le \frac{T}{2} \\ q\_ini + \frac{amax\ T^2}{4} - \frac{amax\ (T-t)^2}{2} & \frac{T}{2} < t \le T \end{cases} \tag{48}$$

```
> q1_profile:=subs(T=Tmin_sys,amax=3,q_ini=0,base_profile);
```

$$q1\_profile := \begin{cases} \frac{3\ t^2}{2} & 0 \le t \le 0.7236012545 \\ 1.570796326 - \frac{3\ (1.447202509 - t)^2}{2} & 0.7236012545 < t \le 1.447202509 \end{cases} \tag{49}$$

```
> q2_profile:=subs(T=Tmin_sys,amax=amax2_res,q_ini=1,base_profile);
```

$$q2\_profile := \begin{cases} 1 + 0.9549296590\ t^2 & 0 \le t \le 0.7236012545 \\ 2.000000000 - 0.9549296590\ (1.447202509 - t)^2 & 0.7236012545 < t \le 1.4472025 \end{cases}$$

```
> q3_profile:=subs(T=Tmin_sys,amax=amax3_res,q_ini=1,base_profile);
```

$$q3\_profile := \begin{cases} 1 + 0.9549296590\ t^2 & 0 \le t \le 0.7236012545 \\ 2.000000000 - 0.9549296590\ (1.447202509 - t)^2 & 0.7236012545 < t \le 1.4472025 \end{cases}$$

```
> plot(q1_profile,t=0..Tmin_sys);
```




```
> plot(q2_profile,t=0..Tmin_sys);
```




```
> plot(q3_profile,t=0..Tmin_sys);
```