```
> restart:with(LinearAlgebra):
```

Define the reference frames undere DH convention






```
> o0:=<0,0,0>;
```

$$o0 := \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix} \tag{1}$$

```
> Mrotztransl:=(theta,point)->><<cos(theta),sin(theta),0,0>|<-sin(theta),
  cos(theta),0,0>|<0,0,1,0>|<point[1],point[2],point[3],1>>;
```

$$Mrotztransl := (\theta,\ point) \mapsto \left\langle \langle \cos(\theta),\ \sin(\theta),\ 0,\ 0\rangle \middle| \langle -\sin(\theta),\ \cos(\theta),\ 0,\ 0\rangle \middle| \langle 0,\ 0,\ 1,\ 0\rangle \middle| \langle point_1,\ point_2,\ point_3,\ 1\rangle \right\rangle \tag{2}$$

```
> Mrotxtransl:=(alpha,point)->><<1,0,0,0>|<0,cos(alpha),sin(alpha),0>|<0,-
```

```
  sin(alpha),cos(alpha),0>|<point[1],point[2],point[3],1>>;
```

$$Mrotxtransl := (\alpha,\ point) \mapsto \left\langle \langle 1,\ 0,\ 0,\ 0\rangle \middle| \langle 0,\ \cos(\alpha),\ \sin(\alpha),\ 0\rangle \middle| \langle 0,\ -\sin(\alpha),\ \cos(\alpha),\ 0\rangle \middle| \langle point_1,\ point_2,\ point_3,\ 1\rangle \right\rangle \tag{3}$$

```
> M01:=Mrotztransl(0,<0,0,q1(t)>).Mrotxtransl(0,<L1,0,0>);
```

$$M01 := \begin{bmatrix} 1 & 0 & 0 & L1 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & q1(t) \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{4}$$

```
> M12:=simplify(Mrotztransl(q2(t),<0,0,0>).Mrotxtransl(0,<L2,0,0>));
```

$$M12 := \begin{bmatrix} \cos(q2(t)) & -\sin(q2(t)) & 0 & \cos(q2(t))\ L2 \\ \sin(q2(t)) & \cos(q2(t)) & 0 & \sin(q2(t))\ L2 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{5}$$

```
> M02:=simplify(M01.M12);
```

$$M02 := \begin{bmatrix} \cos(q2(t)) & -\sin(q2(t)) & 0 & \cos(q2(t))\ L2 + L1 \\ \sin(q2(t)) & \cos(q2(t)) & 0 & \sin(q2(t))\ L2 \\ 0 & 0 & 1 & q1(t) \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{6}$$

```
> E_2:=<0,0,0,1>;
```

$$E\_2 := \begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix} \tag{7}$$

```
> E:=M02.E_2;
```

$$E := \begin{bmatrix} \cos(q2(t))\ L2 + L1 \\ \sin(q2(t))\ L2 \\ q1(t) \\ 1 \end{bmatrix} \tag{8}$$

```
> W_func:=(M)->simplify(map(diff,M,t).MatrixInverse(M));
```

$$W\_func := M \mapsto simplify(map(diff,\ M,\ t) \cdot LinearAlgebra:-MatrixInverse(M)) \tag{9}$$

```
> H_func:=(M)->simplify(map(diff,M,t,t).MatrixInverse(M));
```

$$H\_func := M \mapsto simplify(map(diff,\ M,\ t,\ t) \cdot LinearAlgebra:-MatrixInverse(M)) \tag{10}$$

```
> W01:=W_func(M01);
```

$$W01 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{d}{dt} q1(t) \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{11}$$

```
> Lrot:=<<0, 1, 0, 0>|<-1, 0, 0, 0>|<0, 0, 0, 0>|<0, 0, 0, 0>>;
```

$$Lrot := \begin{bmatrix} 0 & -1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{12}$$

```
> W12:=Lrot*diff(q2(t), t);
```

$$W12 := \begin{bmatrix} 0 & -\frac{d}{dt} q2(t) & 0 & 0 \\ \frac{d}{dt} q2(t) & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{13}$$

```
> W02:=W_func(M02);
```

$$W02 := \begin{bmatrix} 0 & -\frac{d}{dt} q2(t) & 0 & 0 \\ \frac{d}{dt} q2(t) & 0 & 0 & -\left(\frac{d}{dt} q2(t)\right) L1 \\ 0 & 0 & 0 & \frac{d}{dt} q1(t) \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{14}$$

```
> H01:=H_func(M01);
```

$$H01 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{d^2}{dt^2} q1(t) \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{15}$$

```
> H02:=H_func(M02);
```

$$H02 := \begin{bmatrix} -\left(\frac{d}{dt}\, q2(t)\right)^2 & -\frac{d^2}{dt^2}\, q2(t) & 0 & L1\left(\frac{d}{dt}\, q2(t)\right)^2 \\ \frac{d^2}{dt^2}\, q2(t) & -\left(\frac{d}{dt}\, q2(t)\right)^2 & 0 & -L1\left(\frac{d^2}{dt^2}\, q2(t)\right) \\ 0 & 0 & 0 & \frac{d^2}{dt^2}\, q1(t) \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{16}$$

```
> data:={L1=0.5,L2=0.5,m1=10,m2=5,q11=0,q21=-Pi/2,q12=0.5,q22=Pi/2,amaxP=
  5,amaxR=2,g=9.81};
```

$$data := \left\{L1 = 0.5,\ L2 = 0.5,\ amaxP = 5,\ amaxR = 2,\ g = 9.81,\ m1 = 10,\ m2 = 5,\ q11 = 0,\ q12 = 0.5,\ q21 = -\frac{\pi}{2},\ q22 = \frac{\pi}{2}\right\} \tag{17}$$

```
> Tmin:=sqrt(4*abs(deltaq)/amax);
```

$$Tmin := 2\sqrt{\frac{|deltaq|}{amax}} \tag{18}$$

```
> amax_rescaled:=4*deltaq/(T^2);
```

$$amax\_rescaled := \frac{4\ deltaq}{T^2} \tag{19}$$

```
> Tmin_q1:=evalf(subs(deltaq=q12-q11,amax=amaxP,data,Tmin));
```

$$Tmin\_q1 := 0.6324555320 \tag{20}$$

```
> Tmin_q2:=evalf(subs(deltaq=q22-q21,amax=amaxR,data,Tmin));
```

$$Tmin\_q2 := 2.506628274 \tag{21}$$

```
> Tmin_sys:=Tmin_q2;
```

$$Tmin\_sys := 2.506628274 \tag{22}$$

```
> amax1_re:=evalf(subs(T=Tmin_sys,deltaq=q12-q11,data,amax_rescaled));
```

$$amax1\_re := 0.3183098864 \tag{23}$$

```
> amax2_re:=2;
```

$$amax2\_re := 2 \tag{24}$$

```
> base_profile:=piecewise(t>=0 and t<=T/2,qini+1/2*amax*(t^2),t>T/2 and
  t<=T,qini+1/4*amax*(T^2)-1/2*amax*((T-t)^2));
```

$$base\_profile := \begin{cases} qini + \frac{amax\ t^2}{2} & 0 \le t \le \frac{T}{2} \\ qini + \frac{amax\ T^2}{4} - \frac{amax\ (T-t)^2}{2} & \frac{T}{2} < t \le T \end{cases} \tag{25}$$

```
> q1_profile:=subs(qini=q11,amax=amax1_re,T=Tmin_sys,data,base_profile);
```

$$q1\_profile := \begin{cases} 0.1591549432\ t^2 & 0 \le t \le 1.253314137 \\ 0.5000000001 - 0.1591549432\ (2.506628274 - t)^2 & 1.253314137 < t \le 2.5066282 \end{cases}$$

```
> q2_profile:=subs(qini=q21,amax=amax2_re,T=Tmin_sys,data,base_profile);
```

$$q2\_profile := \begin{cases} t^2 - \frac{\pi}{2} & 0 \le t \le 1.253314137 \\ 1.570796325 - (2.506628274 - t)^2 & 1.253314137 < t \le 2.506628274 \end{cases} \quad (27)$$

```
> data
```

$$\{L1 = 0.5,\ L2 = 0.5,\ amaxP = 5,\ amaxR = 2,\ g = 9.81,\ m1 = 10,\ m2 = 5,\ q11 = 0,\ q12 = 0.5,\ q21 = -\frac{\pi}{2},\ q22 = \frac{\pi}{2}\} \quad (28)$$

```
> T_func:=(W,J)->simplify(Trace(1/2*W.J.(W^%T)));
```

$$T\_func := (W, J) \mapsto simplify\left(LinearAlgebra:-Trace\left(\left(\frac{W}{2}\right) \cdot J \cdot W^{T}\right)\right) \quad (29)$$

```
> U_func:=(Hg,J)->simplify(Trace(-Hg.J));
```

$$U\_func := (Hg, J) \mapsto simplify(LinearAlgebra:-Trace(-Hg \cdot J)) \quad (30)$$

```
> J_mobile_to_fixed:=(M,J)->simplify(M.J.(M^%T));
```

$$J\_mobile\_to\_fixed := (M, J) \mapsto simplify\left(M \cdot J \cdot M^{T}\right) \quad (31)$$

```
> I_distributed:=(len,m)->m*(len^2)/3;
```

$$I\_distributed := (len, m) \mapsto \frac{m \cdot len^2}{3} \quad (32)$$

```
> Hg_template:=<<0,0,0,0>|<0,0,0,0>|<0,0,0,0>|<0,0,-g,0>>;
```

$$Hg\_template := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -g \\ 0 & 0 & 0 & 0 \end{bmatrix} \quad (33)$$

```
> G11:=<-L1/2,0,-L1/2>;
```

(34)

$$G11 := \begin{bmatrix} -\frac{L1}{2} \\ 0 \\ -\frac{L1}{2} \end{bmatrix} \tag{34}$$

```
> G22:=<-L2/2,0,0>;
```

$$G22 := \begin{bmatrix} -\frac{L2}{2} \\ 0 \\ 0 \end{bmatrix} \tag{35}$$

```
> J11:=<<0,0,0,m1*G11[1]>|<0,0,0,m1*G11[2]>|<0,0,0,m1*G11[3]>|<m1*G11[1],
  m1*G11[2],m1*G11[3],m1>>;
```

$$J11 := \begin{bmatrix} 0 & 0 & 0 & -\frac{m1\ L1}{2} \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -\frac{m1\ L1}{2} \\ -\frac{m1\ L1}{2} & 0 & -\frac{m1\ L1}{2} & m1 \end{bmatrix} \tag{36}$$

```
> J22:=<<I_distributed(-L2,m2),0,0,m2*G22[1]>|<0,0,0,m2*G22[2]>|<0,0,0,
  m2*G22[3]>|<m2*G22[1],m2*G22[2],m2*G22[3],m2>>;
```

$$J22 := \begin{bmatrix} \frac{m2\ L2^2}{3} & 0 & 0 & -\frac{m2\ L2}{2} \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ -\frac{m2\ L2}{2} & 0 & 0 & m2 \end{bmatrix} \tag{37}$$

```
> J10:=J_mobile_to_fixed(M01,J11);
```

$$J10 := \tag{38}$$

$$\begin{bmatrix} 0 & 0 & -\frac{m1\ L1\ (L1-q1(t))}{2} & \frac{m1\ L1}{2} \\ 0 & 0 & 0 & 0 \\ -\frac{m1\ L1\ (L1-q1(t))}{2} & 0 & -q1(t)\ m1\ (L1-q1(t)) & -\frac{m1\ (L1-2\ q1(t))}{2} \\ \frac{m1\ L1}{2} & 0 & -\frac{m1\ (L1-2\ q1(t))}{2} & m1 \end{bmatrix}$$

```
> J20:=J_mobile_to_fixed(M02,J22);
```

$$J20 := \left[\left[\frac{m2\,(\cos(q2(t))^2\,L2^2 + 3\,\cos(q2(t))\,L1\,L2 + 3\,L1^2)}{3},\right.\right. \tag{39}$$

$$\frac{\sin(q2(t))\,L2\,m2\,(2\,\cos(q2(t))\,L2 + 3\,L1)}{6},$$

$$\left.\frac{q1(t)\,m2\,(\cos(q2(t))\,L2 + 2\,L1)}{2},\ \frac{m2\,(\cos(q2(t))\,L2 + 2\,L1)}{2}\right],$$

$$\left[\frac{\sin(q2(t))\,L2\,m2\,(2\,\cos(q2(t))\,L2 + 3\,L1)}{6},\ \frac{\sin(q2(t))^2\,m2\,L2^2}{3},\right.$$

$$\left.\frac{\sin(q2(t))\,m2\,L2\,q1(t)}{2},\ \frac{\sin(q2(t))\,m2\,L2}{2}\right],$$

$$\left[\frac{q1(t)\,m2\,(\cos(q2(t))\,L2 + 2\,L1)}{2},\ \frac{\sin(q2(t))\,m2\,L2\,q1(t)}{2},\ q1(t)^2\,m2,\right.$$

$$\left.q1(t)\,m2\right],$$

$$\left.\left[\frac{m2\,(\cos(q2(t))\,L2 + 2\,L1)}{2},\ \frac{\sin(q2(t))\,m2\,L2}{2},\ q1(t)\,m2,\ m2\right]\right]$$

```
> T1:=T_func(W01,J10);
```

$$T1 := \frac{\left(\frac{d}{dt}\,q1(t)\right)^2\,m1}{2} \tag{40}$$

```
> T2:=T_func(W02,J20);
```

$$T2 := \frac{m2\left(3\left(\frac{d}{dt}\,q1(t)\right)^2 + L2^2\left(\frac{d}{dt}\,q2(t)\right)^2\right)}{6} \tag{41}$$

```
> Hg1:=Hg_template;
```

$$Hg1 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -g \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{42}$$

```
> Hg2:=Hg_template;
```

$$Hg2 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -g \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{43}$$

```
> U1:=U_func(Hg1,J10);
```

$$U1 := -\frac{g\ m1\ (L1 - 2\ q1(t))}{2} \tag{44}$$

```
> U2:=U_func(Hg2,J20);
```

$$U2 := g\ q1(t)\ m2 \tag{45}$$

```
> Lagr:=T1+T2-U1-U2;
```

$$Lagr := \frac{\left(\frac{d}{dt}\ q1(t)\right)^2\ m1}{2} + \frac{m2\left(3\left(\frac{d}{dt}\ q1(t)\right)^2 + L2^2\left(\frac{d}{dt}\ q2(t)\right)^2\right)}{6} + \frac{g\ m1\ (L1 - 2\ q1(t))}{2} - g\ q1(t)\ m2 \tag{46}$$

```
> diffF := (f,x)->subs(y=x,diff(subs(x=y,f),y));
```

$$diffF := (f,\ x) \rightarrow subs\left(y = x,\ \frac{\partial}{\partial y}\ subs(x = y,\ f)\right) \tag{47}$$

```
> EQM_left:=(Lagr,qt)->simplify(diff(diffF(Lagr,diff(qt,t)),t) - diffF
  (Lagr,qt));
```

$$EQM\_left := (Lagr,\ qt) \rightarrow simplify\left(\frac{\partial}{\partial t}\ diffF\left(Lagr,\ \frac{\partial}{\partial t}\ qt\right) - diffF(Lagr,\ qt)\right) \tag{48}$$

```
> EQM1:=EQM_left(Lagr,q1(t))-F1;
```

$$EQM1 := (m1 + m2)\left(g + \frac{d^2}{dt^2}\ q1(t)\right) - F1 \tag{49}$$

```
> EQM2:=EQM_left(Lagr,q2(t))-C2;
```

$$EQM2 := \frac{m2\ L2^2\left(\frac{d^2}{dt^2}\ q2(t)\right)}{3} - C2 \tag{50}$$

```
> f1:=solve(subs(data,EQM1),F1);
```

$$f1 := 147.1500000 + 15.\ \frac{d^2}{dt^2}\ q1(t) \tag{51}$$

```
> c2:=solve(subs(data,EQM2),C2);
```

$$c2 := 0.4166666667\ \frac{d^2}{dt^2}\ q2(t) \tag{52}$$

```
> f1_profile:=subs(q1(t)=q1_profile,q2(t)=q2_profile,f1);
```

$$f1\_profile := 147.1500000 + 15.\ \frac{d^2}{dt^2}\begin{cases} 0.1591549432\ t^2 & \\ 0.5000000001 - 0.1591549432\ (2.506628274 - t)^2 & 1.2533 \end{cases}$$

```
> c2_profile:=subs(q1(t)=q1_profile,q2(t)=q2_profile,c2);
```

$$c2\_profile := 0.4166666667 \frac{d^2}{dt^2} \begin{cases} t^2 - \frac{\pi}{2} & 0 \le t \le 1.253314137 \\ 1.570796325 - (2.506628274 - t)^2 & 1.253314137 < t \le 2.506628274 \end{cases} \quad (54)$$

```
> plot(f1_profile,t=0..Tmin_sys);
```




```
> plot(c2_profile,t=0..Tmin_sys);
```